EXPLORATIONS

APPENDIX B

Probability Distributions

The four programs given in this Appendix are used to compute P($X \le a$) for the Binomial, Poisson, Standard Normal, and Student-t distributions. We note that the program name for the Normal distribution had to be modified to NRMAL since the TI-86 reserves the name Normal for a mode setting. Applications of the four programs are illustrated below. *Pay particular attention to the limitations of the Binomial and Poisson programs discussed at the end of this Appendix.*

Binomial Distribution

```
BINOMIAL

:ClLCD
:Disp "If X is binomial(n,p)"
:Disp "then this program"
:Disp "finds P(X≤i)."
:Input "n=",N
:Input "p=",P
:Input "i=",I
:(1-P)^N→T
:T→PROB
:If I==Ø
:Goto LABEL
:For(K,Ø,I-1,1)
:(P/(1-P))((N-K)/(K+1))T→T
:PROB+T→PROB
:End
:Lbl LABEL
:Disp "P(X≤i)=",PROB
:Stop
```

If X is Binomial with $n = 1000$ and $p = 0.1$, then (B.1) and (B.2) show that:

P($X \le 90$) = .158238141393.

```
If X is binomial(n,p)
then this program
finds P(X≤i).
n=1000
p=.1
i=90
```

(B.1)

```
finds P(X≤i).
n=1000
p=.1
i=90
P(X≤i)=
          .158238141393
                   Done
```

(B.2)

Poisson Distribution

```
                    POISSON
:ClLCD
:Disp "If X is Poisson(λ)"
:Disp "then this program"
:Disp "finds P(X≤I)."
:Input "λ=",λ
:Input "i=",I
:e^(-λ)→T
:T→PROB
:If I==Ø
:Goto LABEL
:For(K,Ø,I-1,1)
:(λ/(K+1))T→T
:PROB+T→PROB
:End
:Lbl LABEL
:Disp "P(X≤i)=",PROB
:Stop
```

If X is Poisson with $\lambda = 100$, then (B.3) and (B.4) show that

$P(X \leq 90) = .171385119322.$



(B.3)



(B.4)

Standard Normal Distribution

```
                    NRMAL
:ClLCD
:.ØØØØ1→tol
:Disp "If X is normal(Ø,1)"
:Disp "then this program"
:Disp "finds φ(a)=P(X≤a)."
:Input "a=",A
:abs A→A1
:.5+(1/√(2π))fnInt(e^(-x² /2),x,Ø,A1)→PROB
:If A<Ø
:1-PROB→PROB
:Disp "φ(a)=",PROB
:Stop
```

If X is Standard Normal, then (B.5) and (B.6) show that

$P(X \leq (5-3)/3) = .747507462453.$



(B.5)



(B.6)

Further, (B.7) and (B.8) show that $P(X \leq (1-3)/3) = .252492537547.$



(B.7)



(B.8)

Student-*t* Distributions

```
STUt
:ClLCD
:.00001→tol
:Disp "If X is Student t"
:Disp "with df=n then this"
:Disp "program finds P(X≤a)."
:Input "n=",N
:Input "a=",A
:abs A→A1
:1/√π→C
:If N==1
:Goto LABEL
:For(K,1,N-1)
:K/(2C)→C
:End
:Lbl LABEL
:.5+C(1/√(Nπ))fnInt((1+x² /N)^(⁻.5(N+1)),x,
0,A1)→PROB
:If A<0
:1-PROB→PROB
:Disp "P(X≤a)=",PROB
:Stop
```

If X is Student-t with 15 degrees of freedom then (B.9) and (B.10) show that

$P(X \le 1.34) = .899903868291$, and

```
If X is Student t
with df=n then this
program finds P(X≤a).
n=15
a=1.34
```

(B.9)

```
with df=n then this
program finds P(X≤a).
n=15
a=1.34
P(X≤a)=
         .899903868291
                  Done
```

(B.10)

Further, (B.11) and (B.12) show that $P(X \le -1.34) = .100096131709$.

```
If X is Student t
with df=n then this
program finds P(X≤a).
n=15
a=⁻1.34
```

(B.11)

```
with df=n then this
program finds P(X≤a).
n=15
a=⁻1.34
P(X≤a)=
         .100096131709
                  Done
```

(B.12)

Numerical Limitations of the Binomial and Poisson Programs

Binomial Program

The Binomial program fails when n and p are such that the TI-86 thinks $(1 - p)^n$ is 0. For such values of n and p, the Binomial program gives $P(X \le i) = 0$ for all i.

For example, as (B.13) shows, when $n = 3319$ and $p = 1/2$, the TI-86 thinks $(1 - p)^n$ is 0.

```
(1-1/2)^3319
                     0
(1-1/2)^3317
   3.04441871186E-999
```

(B.13)

Thus, as shown in (B.14) and (B.15) the Binomial program gives P($X \le 1659$) = 0, instead of the correct value P($X \le 1659$) = 1/2.

```
If X is binomial(n,p)
then this program
finds P(X≤i).
n=3319
p=1/2
i=1659
```

(B.14)

```
finds P(X≤i).
n=3319
p=1/2
i=1659
P(X≤i)=
                      0
                   Done
```

(B.15)

On the other hand, as (B.13) shows when $n = 3317$ and $p = 1/2$, the TI-86 does not think $(1-p)^n$ is 0. In this case, as shown in (B.16) and (B.17), the Binomial program gives the correct result P($X \le 1658$) = 1/2.

```
If X is binomial(n,p)
then this program
finds P(X≤i).
n=3317
p=1/2
i=1658
```

(B.16)

```
finds P(X≤i).
n=3317
p=1/2
i=1658
P(X≤i)=
                     .5
                   Done
```

(B.17)

Poisson Program

The Poisson program has a similar limitation. It fails when $\lambda > 0$ is chosen, so that the TI-86 thinks $e^{-\lambda}$ is 0. The largest positive integer λ for which $e^{-\lambda}$ is not considered 0 by the TI-86 is $\lambda = 2300$. See (B.18).

```
e^-2300
    1.32645225542E-999
e^-2301
                      0
```

(B.18)

Thus, do not use the Poisson program for $\lambda > 2300$.

When the TI-86 thinks $e^{-\lambda}$ is 0, then, as in the Binomial program, it gives P($X \le i$) = 0 for all i.

